Inference control and meta-programming

Lecture 4

Inference control and meta-programming

(in PROLOG)

- Negation as Failure
- Inference control
- Cut to control backtracking
- Meta-predicates (Any Prolog textbook e.g. SS)
- Terms and their representation
- Meta-interpreters

SLDNF = SLD-resolution + NF

NF (Negation as Finite Failure):
an atom $\neg A$ succeeds if the derivation tree corresponding to the goal A is finite and its leaves are all failure leaves.

Consistent (with the various characterizations of negation), but incomplete.

Conditions for completeness are very stringent:

- instantiated variables in negated atoms
- constraints on the program structure

Answer Set Programming extends SLDNF, computing the answer on the basis of the stable model semantics

Negation as Failure in PROLOG

In prolog negation is realized as **failure** in the search.

If `X` is a list of atoms, `not(X)` is true if the interpreter cannot find a proof for `X`.

```
student(bill).
student(joe).
married(joe).

unmarried_student(X) :- not(married(X)), student(X).
? unmarried_student(bill/joe).
```

Termination of negation

The negation mechanism of Prolog is neither correct nor complete: it depends on the ordering of evaluation of clauses

The termination of `not(X)` depends on the termination of `X`:
- If `X` terminates, `not(X)` terminates.
- If `X` has infinite solutions, `not(X)` terminates if a success node is found before an infinite branch is encountered.

```
married(gino, remberta).
married(X,Y):-married(Y,X).

? not(married(remberta, gino)).
```

Example

```
p(s(X)):-p(X).
q(a).
```

The query `not(p(X),q(X))` does not terminate.

Instead, `not(q(X),p(X))` terminates.

Using negation with completely instantiated atoms does not generate incorrect behaviors, in case the atoms terminate. If negation is used with variables we must take into account the execution mechanism of prolog

Problems with negation

```
unmarried_student(X):-not(married(X)),student(X).
student(bill).
married(joe).
?- unmarried_student(X).
```

```
unmarried_student1(X) :- student(X), not(married(X)).
student(bill).
student(joe).
married(joe).
?- unmarried_student1(X).
```

Sufficient condition for correctness:
variables in atoms must be instantiated before they are executed

Inference control in PROLOG programs

- Program specification
- Clause ordering
- Conjunct ordering
- Cut to control backtracking

PROLOG program specification

```
descendant(X,Y):-son(X,Y). % 1
descendant(X,Y):-son(Z,Y),descendant(X,Z).

descendant(X,Y):-son(X,Y). % 2
descendant(X,Y):-son(X,Z),descendant(Z,Y).

descendant(X,Y):-son(X,Y). % 2
descendant(X,Y):-descendant(X,Z),descendant(Z,Y).
```

Explicit Inference Control

"What is the income of the US president' wife?"

```
income(S,I),married(S,P),job(P,uspresident).
```

"Do I have an american cousin?"

```
american(Y),cousin(daniele,Y).
```

♢ **meta-reasoning** to decide which order of the atoms is more effective

Given a predicate returning the number of instances of a relation, one can sort the conjuncts to optimize the computation of the answer.

Efficiency of PROLOG implementations

- **chronological backtracking**
- **dependency directed backtracking**
- **intelligent backtracking** memorizing the previous derivation to avoid re-discovering them

Cut

Generic clause with cut:

$$A \leftarrow B_1, \ldots, B_k, !, B_{k+2}, B_n.$$

If the current goal G unifies with A and $B_1, \ldots, B_k$ are successfully proven,

- any other proof achievable by unifying G with another clause is eliminated from the search tree;
- any alternative proof of $B_1, \ldots, B_k$, is also eliminated from the search tree.

Cut: example

```
merge1([X|Xs],[Y|Ys],[X|Zs]):-
        X<Y,!,
        merge1(Xs,[Y|Ys],Zs).
merge1([X|Xs],[Y|Ys],[X,Y|Zs]):-
    X=:=Y,!,
    merge1(Xs,Ys,Zs).
merge1([X|Xs],[Y|Ys],[Y|Zs]):-
    X>Y,!,
    merge1([X|Xs],Ys,Zs).
merge1([],X,X):- !.
merge1(Y,[],Y):- !.
```

Green and red cuts: discarding valid solutions

A cut is **green**, when it does not affect the program solutions, **red** otherwise.

```
ancestor(X,Y):- parent(X,Y).
ancestor(X,Y):- ancestor(X,Z),!,parent(Z,Y).
```

Stops the search at the first ancestor of X: Z. Since Y may not be parent of Z, the search fails.

Controlling backtracking

```
cousin(massimo,daniele),!,american(daniele)

member(a,C), !, q(a).

member1(X,[X|Xs]) :- !.
member1(X,[Y|Ys]) :- member1(X,Ys).

member1(a,C).
```

Meta-logical predicates

♢ Needed to deal with a program element as data.

♢ Access the program's internal representation.

♢ Enable meta-programming.

♢ Meta-programming makes it possible to **explicitly** control execution

Collecting solutions

It is often useful to compute all the solutions of a program.

`findall(Term,Goal,List).`

`List` is the list, obtained computing `Goal` and taking the bindings of `Term`.

`findall(X,grandfather(giovanni,X),Grandchildren).`

Variables as programs

- `call(X)` executes the goal `X` which must be instantiated, (simply written `X`).

Es: Disjunction

```
X ; Y :- X.
X ; Y :- Y.
```

Negation as failure

```
not(G):- G, !, fail.
not(G).
```

`fail` always fails.

Type meta-logical predicates

- `var(Term)` succeeds if `Term` is a variable
- `nonvar(Term)` succeeds if `Term` is not a variable

```
grandfather(X,Z):- nonvar(X),
                   parent(X,Y),
                   parent(Y,Z).
grandfather(X,Z):- nonvar(Z),
                   parent(Y,Z),
                   parent(X,Y).
```

Ground terms

```
ground(Term) :- nonvar(Term), constant(Term).
ground(Term) :-  nonvar(Term), compound(Term),
                functor(Term,F,N), ground(N,Term).

ground(N,Term) :- N > 0, arg(N,Term,Arg),
                  ground(Arg), N1 is N-1,
                  ground(N1,Term).
ground(0,Term).
```

Equality and Unification

- `X == Y` succeeds if `X` and `Y` are the same constant, the same variable, or structures with the same functor and recursively == (negated \ ==)
- `X = Y` succeeds if `X` and `Y` unify

Unification

```
unify1(X,Y) :- var(X), var(Y), X=Y.
unify1(X,Y) :- var(X), nonvar(Y),
               not_occurs_in(X,Y), X=Y.
unify1(X,Y) :- var(Y), nonvar(X),
               not_occurs_in(Y,X), Y=X.
unify1(X,Y) :- nonvar(X), nonvar(Y),
               constant(X), constant(Y), X=Y.
unify1(X,Y) :- nonvar(X), nonvar(Y), compound(X),
               compound(Y), term_unify1(X,Y).

term_unify1(X,Y) :- functor(X,F,N), functor(Y,F,N),
                    unify1_args(N,X,Y).
% Program 10.6    Unification with the occurs check
```

Argument unification

```
unify1_args(N,X,Y) :- N > 0, unify1_arg(N,X,Y),
                      N1 is N-1, unify1_args(N1,X,Y).
unify1_args(0,X,Y).

unify1_arg(N,X,Y) :- arg(N,X,ArgX), arg(N,Y,ArgY),
                     unify1(ArgX,ArgY).
```

Occurs check

```
not_occurs_in(X,Y) :- var(Y), X \== Y.
not_occurs_in(X,Y) :- nonvar(Y), constant(Y).
not_occurs_in(X,Y) :- nonvar(Y), compound(Y),
                      functor(Y,F,N),
                      not_occurs_in(N,X,Y).

not_occurs_in(N,X,Y) :- N > 0, arg(N,Y,Arg),
                        not_occurs_in(X,Arg),
                        N1 is N-1,
                        not_occurs_in(N1,X,Y).
not_occurs_in(0,X,Y).
```

Basic meta interpreter

A PROLOG **meta-interpreter** is a PROLOG interpreter written in PROLOG.

```
solve1(true).
solve1((A,B)) :- solve1(A), solve1(B).
solve1(A) :- clause(A,B), solve1(B).
```

`clause/2` returns the head and the body of the clause (at least one arg must be instantiated).

Meta interpreter with unification

```
solve2(true).
solve2((A,B)) :- solve2(A), solve2(B).
solve2(A) :- clause(X,B), unify1(X,A), solve2(B).
```

Meta interpreter tracing the proof

```
solve_trace(Goal) :- solve_trace(Goal,0).

solve_trace(true,Depth) :- !.
solve_trace((A,B),Depth) :-  !, solve_trace(A,Depth),
                             solve_trace(B,Depth).
solve_trace(A,Depth) :- builtin(A), !, A,
                        display(A,Depth), nl.
solve_trace(A,Depth) :- clause(A,B), display(A,Depth),
                        nl, Depth1 is Depth + 1,
                        solve_trace(B,Depth1).

display(A,Depth) :-  Spacing is 3*Depth,
                     put_spaces(Spacing), write(A).
```

```
put_spaces(N) :- between(1,N,I), put_char(' '), fail.
put_spaces(N).

between(I,J,I) :- I =< J.
between(I,J,K) :- I < J, I1 is I + 1, between(I1,J,K).

% Program 17.7 A tracer for Prolog
```

Meta interpreter building the proof tree

```
solve(true,true) :- !.
solve((A,B),(ProofA,ProofB)) :- !, solve(A,ProofA),
                                 solve(B,ProofB).
solve(A,(A:-builtin)) :- builtin(A), !, A.
solve(A,(A:-Proof)) :- clause(A,B), solve(B,Proof).

% Program 17.8
% A meta-interpreter for building a proof tree
```

Meta interpreter with certainty factors

```
solve(true,1) :- !.
solve((A,B),C) :- !, solve(A,C1),
                     solve(B,C2),
                     minimum(C1,C2,C).
solve(A,1) :- builtin(A), !, A.
solve(A,C) :-clause_cf(A,B,C1),
                solve(B,C2), C is C1 * C2.

minimum(X,Y,X) :- X =< Y, !.
minimum(X,Y,Y) :- X > Y, !.

%  Program 17.9
% A meta-interpreter for reasoning with uncertainty
```

Esercises (i)

- Define in PROLOG the relation `onlychild(X)`, exploiting the family defined before.
- Define the predicate `notMember(X,L)`, true if `X` does not occur in the list `L`. Provide a definition using NAF and one without it, and compare them.
- Define union using the negation of `member`. Build the serach tree for the goal
  `? union([a,b],[b,c],Z).`

Esercises (ii)

- Apply the cut operator to the program `insert` of assignment 2, and draw the search tree for the goal `insert(4, [3,5,7], X)`.
- Define a predicate `sortm(X,Y,Z)`, to order lists `X` and `Y` and return the result in `Z`, through merge-sort.
- Add cuts to the program `sortm` if needed to obtain a single solution.

Esercises (iii)

- Check the execution of the vanilla meta-interpreter by computing queries of predicates `grandfather`, `descendant` of the family program.
- Consider the vanilla meta-interpreter returning the proof tree. Add a few built-in predicates and verify the behaviour of the meta interpreter on built-in and user-defined `member`. Compare the proof-tree obtained with the meta-interpreter tracing the execution of the same query with the standard interpreter.